SIGMA16

COMPARISON INSTRUCTIONS

cmp compare

cmplt compare **less than**

cmpeq compare **equal**

cmpgt compare **greater than**

CONDITIONAL JUMPS

jump jump

jumpf jump if **false**

jumpt jump if **true**

IF THEN ELSE

```
if x < y:
    x = 3
else:
    x = 6
y = 5
```

```
    load    R1,x[R0]    ; R1 := x
    load    R2,y[R0]    ; R2 := y
    cmplt   R3,R1,R2    ; R3 := x<y
    jumpf   R3,else[R0] ; if x >= y, goto else
    lea     R1,3[R0]    ; R1 := 3
    jump    done[R0]    ; skip else
else
    lea     R1,6[R0]    ; R1 := 6
done
    lea     R2,5[R0]    ; R2 := 5
    store   R1,x[R0]
    store   R2,y[R0]

x   data    10
y   data    15
```

WHILE LOOP

```
while x < y:
    x += 1
y += x
```

```
    load    R1,x[R0]    ; R1 := x
    load    R2,y[R0]    ; R2 := y
    lea     R3,1[R0]    ; R3 := 1
loop
    cmplt   R4,R1,R2    ; R3 := x<y
    jumpf   R4,done[R0] ; if x >= y, exit loop
    add     R1,R1,R3    ; R1 := R1 + 1
    jump    loop[R0]    ; repeat loop
done
    add     R2,R2,R1    ; R2 := R2 + R1
    store   R1,x[R0]
    store   R2,y[R0]

x   data    1
y   data    10
```

FOR LOOP

```
for i in range(5):
    x += x
```

```
    lea     R1,0[R0]    ; i = 0
    lea     R2,5[R0]    ; temp = 5
    lea     R3,1[R0]    ; R3 := const 1
    load    R4,x[R0]    ; R4 := x
loop
    cmplt   R5,R1,R2    ; R4 := i < temp
    jumpf   R5,done[R0] ; if i >= 5, exit loop
    add     R4,R4,R4    ; x += x
    add     R1,R1,R3    ; i += 1
    jump    loop[R0]    ; repeat loop
done
    store   R4,x[R0]

x   data    2
```

ARRAYS

```
n   data    5   ; size of array x

x   data    13  ; x[0]
    data    120 ; x[1]
    data    100 ; x[2]
    data    0   ; x[3]
    data    37  ; x[4]
```

RECORDS

```
    load    R1,x_fieldB[R0] ; R1 := x.fieldB
    load    R2,x_fieldC[R0] ; R2 := x.fieldC
    add     R1,R1,R2        ; x.fieldA = x.fieldB + x.fieldC
    store   R1,x_fieldA[R0]

; Record x
x
x_fieldA    data    3
x_fieldB    data    4
x_fieldC    data    5

; Record y
y
y_fieldA    data    20
y_fieldB    data    21
y_fieldC    data    22
```